15.排程浮點數程式碼 (PPlain和PMMX)





    浮點數指令無法像整數指令一樣組合在一起(pair)﹐除了一種定義於下的特別情況外﹕





第一個指令(在U-pipe中執行)必須是FLD, FADD, FSUB, FMUL, FDIV, FCOM, FCHS, 或FABS。





第二個指令必須是(在V-pipe中)FXCH。





在FXCH之後的指令必須是一個浮點數指令﹐否則FXCH將成對(pair)不完全而且需要一個額外的時計週期。





這種特別的組成是重要的﹐下面會有簡短的解釋。





雖然浮點數指令一般是無法組成的﹐但很多是可以管路化的﹐這就是說﹐一個指令可以開始在前一個指令結束前﹐例如﹕





FADD ST(1),ST(0)   ; clock cycle 1-3


FADD ST(2),ST(0)   ; clock cycle 2-4


FADD ST(3),ST(0)   ; clock cycle 3-5


FADD ST(4),ST(0)   ; clock cycle 4-6





    顯然﹐兩個指令無法交疊(overlap)如果第二個指令需要第一個指令的結果。既然幾乎所有的浮點數指令都與堆疊暫存器的頂端﹐ST(0)有關﹐那麼就不太可能將一個指令獨立於前一個指令的結果。這個問題的解決方法是暫存器的重新命名。FXCH指令實際上並沒有將兩個暫存器中的內容交換﹐它只交換了它們的名字。壓入及挽出堆疊的指令也是以重新命名在執行的。浮點數暫存器重新命名已經在Pentium上被高度最佳化了﹐以至於一個暫存器可能在使用中的時後被重新命名。暫存器重新命名從未造成拖延(stall)-甚至想在同一個時計週期中重新命名一個暫存器一次以上也是可能的﹐例如﹐當你將FXCH與FLD或FCOMPP組成的時候。





藉著是當的使用FXCH指令﹐你可以在妳的浮點數程式碼中得到很多的交疊﹐例如﹕





FLD     [a1]    ; clock cycle 1


FADD    [a2]    ; clock cycle 2-4


FLD     [b1]    ; clock cycle 3


FADD    [b2]    ; clock cycle 4-6


FLD     [c1]    ; clock cycle 5


FADD    [c2]    ; clock cycle 6-8


FXCH    ST(2)   ; clock cycle 6


FADD    [a3]    ; clock cycle 7-9


FXCH    ST(1)   ; clock cycle 7


FADD    [b3]    ; clock cycle 8-10


FXCH    ST(2)   ; clock cycle 8


FADD    [c3]    ; clock cycle 9-11


FXCH    ST(1)   ; clock cycle 9


FADD    [a4]    ; clock cycle 10-12


FXCH    ST(2)   ; clock cycle 10


FADD    [b4]    ; clock cycle 11-13


FXCH    ST(1)   ; clock cycle 11


FADD    [c4]    ; clock cycle 12-14


FXCH    ST(2)   ; clock cycle 12





   在上面的例子中我們插入了三個獨立的串列(thread)。每一個FADD需要三個時計週期﹐而我們可以在每一個時計週期中開始一個新的FADD。當我們在a串列中開始一個FADD我們有時間在a傳回之前在b和c串列中去開始兩個新的FADD指令﹐所以每三個FADD屬於同一個串列。我們每次都用FXCH指令將所要串列中的暫存器放入ST(0)中。如你在上面的例子中所見﹐這樣就產生了一個常規模式(regular pattern)﹐但注意FXCH每兩個就重複﹐而串列每三個就重複。這可以變得很令人困惑﹐所以為了知道哪個暫存器在哪，你必須去"玩電腦"。





   所有版本的FADD﹐FSUB﹐FMUL和FILD都需要三個時計週期而且可以交疊﹐以至於這些指令可以用上面所說的方法來排程。使用記憶體運算元不會比使用暫存器運算元花更多的時間﹐如果記憶體運算元是在第一級cache(level 1 cache)而且適當的對齊(align)的話。





    現在你一定很習慣於有例外的規則了﹐但交疊是沒有例外的﹕你不能在一個FMUL開始後的一個時計週期開始另一個FMUL﹐因為FMUL的電器迴路沒有完美得管路化。建議你在每兩FMUL中放入另一個指令。例如﹕





FLD     [a1]    ; clock cycle 1


FLD     [b1]    ; clock cycle 2


FLD     [c1]    ; clock cycle 3


FXCH    ST(2)   ; clock cycle 3


FMUL    [a2]    ; clock cycle 4-6


FXCH            ; clock cycle 4


FMUL    [b2]    ; clock cycle 5-7    (stall)


FXCH    ST(2)   ; clock cycle 5


FMUL    [c2]    ; clock cycle 7-9    (stall)


FXCH            ; clock cycle 7


FSTP    [a3]    ; clock cycle 8-9


FXCH            ; clock cycle 10     (unpaired)


FSTP    [b3]    ; clock cycle 11-12


FSTP    [c3]    ; clock cycle 13-14





    在這﹐你在FMUL[b2]和FMUL[c2]之前有一個拖延(stall)﹐因為在之前的一個時計週期中開始了另一個FMUL。你可以在FMUL中放入FLD指令來改進這個程式碼。





FLD     [a1]    ; clock cycle 1


FMUL    [a2]    ; clock cycle 2-4


FLD	     [b1]    ; clock cycle 3


FMU    [b2]    ; clock cycle 4-6


FLD     [c1]    ; clock cycle 5


FMUL    [c2]    ; clock cycle 6-8


FXCH    ST(2)   ; clock cycle 6


FSTP    [a3]    ; clock cycle 7-8


FSTP    [b3]    ; clock cycle 9-10


FSTP    [c3]    ; clock cycle 11-12





    情況中你可以放入FADD﹐FSUB會任何其他的指令在FMUL之中來避免拖延。





    交疊點數指令需要一些讓你能插入的獨立串列。如果你只有一個大式子要執行﹐你可以平行計算部分的式子來達到交疊的目的。例如﹐如果你有六個數要加﹐你可以將運算分成兩串列﹐三個數一個串列﹐然後最後將兩個串列相加﹕





FLD     [a]     ; clock cycle 1


FADD    [b]     ; clock cycle 2-4


FLD     [c]     ; clock cycle 3


FADD    [d]     ; clock cycle 4-6


FXCH            ; clock cycle 4


FADD    [e]     ; clock cycle 5-7


FXCH            ; clock cycle 5


FADD    [f]     ; clock cycle 7-9    (stall)


FADD            ; clock cycle 10-12  (stall)





    在這﹐在FADD[f]前有一個時計拖延因為它在等FADD[d]的結果﹐在最後一個FADD前有兩個時計拖延因為它在等FADD[f]的結果。第二個拖延可以藉由加入一個整數指令而藏起來﹐但第一個卻不行﹐因為在這插入一個整數指令會讓FXCH無法組成。





    第一個拖延可以藉由三個串列而避免﹐但這將會花費多一個FLD所以我們並沒有省了什麼時間--除非我們至少有8個數要加。


 


    不是所有的浮點數指令都可以交疊。而有些浮點數指令能比交疊浮點數指令次序列交疊更多整數指令次序列。舉例來說﹐FDIV指令需要39個時計週期。除了第一個時計週期外都能用整數指令交疊﹐但只有最後兩個時計週期能用浮點數指令週期。例如﹕





FDIV            ; clock cycle 1-39  (U-pipe)


FXCH            ; clock cycle 1-2   (V-pipe, imperfect pairing)


SHR EAX,1       ; clock cycle 3     (U-pipe)


INC EBX         ; clock cycle 3     (V-pipe)


CMC             ; clock cycle 4-5   (non-pairable)


FADD [x]        ; clock cycle 38-40 (U-pipe, waiting while f.p. unit busy)


FXCH            ; clock cycle 38    (V-pipe)


FMUL [y]        ; clock cycle 40-42 (U-pipe, waiting for result of FDIV)


	


    第一個DXCH用FDIV組成﹐但需要一個額外的時計週期，因為它並不是在一個浮點數指令之前。SHR/INC組開始在FDIV結束前﹐但必須等FXCH結束。FADD必須要等到第38時計﹐因為新的浮點數指令只能在FDIV的最後兩個時計週期執行。第二個FXCH用FADD組成。FMUL必須等FDIV結束因為它用到除法的結果。





    有一大個整數交疊可以放在浮點數指令後面﹐例如﹕FDIV或FSQRT﹐的話﹐那麼你可以放一個虛擬指令來讀取位址﹐這個指令可以是你之後會要用到﹐只是先確定它有放在第一級cache中而已。例如﹕





        FDIV    QWORD PTR [EBX]


        CMP     [ESI],EAX


        FMUL    QWORD PTR [ESI]





    們用了整數交疊先讀取了[ESI]到cache中當FDIV在執行的時候。（我們並不在乎CMP的結果是什麼。）





    21節提供了一個完整的浮點數指令表列﹐和他們可以用什麼來組成或交疊。





    在浮點數指令上用記憶體運算元並不會有什麼不好﹐因為算數運算單元總是落後讀取單元一步。這種交換的益處發生在當你把浮點數檔案存入記憶體中時。使用記憶體運算元的FST或FSTP指令需要兩個時計週期來執行﹐但它在之前一個時計就需要檔案﹐所以你會得到一個時計的拖延如果在前一個時計檔案並沒有存入的話。這對於ASI拖延是類似的。例如﹕





FLD     [a1]    ; clock cycle 1


FADD    [a2]    ; clock cycle 2-4


FLD     [b1]    ; clock cycle 3


FADD    [b2]    ; clock cycle 4-6


FXCH            ; clock cycle 4


FSTP    [a3]    ; clock cycle 6-7


FSTP    [b3]    ; clock cycle 8-9


	


FSTP[a3]拖延了一個時計週期，因為FADD[a2]的結果並沒有在前一個時計週期就準備好。在很多情況中﹐你無法影藏這種拖延而不用把你的浮點數程式碼排成成4個串列﹐或者在之間放入幾個整數指令。





    在FST(P)指令執行中的兩個時計週期是無法用任何次序列指令組成或交疊的。





    使用整數運算元的指令如﹕FIADD﹐FISUB﹐FIMUL﹐FIDIV﹐FICOM可以被分裂成更簡單的運算﹐為了改進交疊。例如﹕





FILD    [a]     ; clock cycle 1-3


FIMUL   [b]     ; clock cycle 4-9





分裂成﹕


FILD    [a]     ; clock cycle 1-3


FILD    [b]     ; clock cycle 2-4


FMUL            ; clock cycle 5-7





在這個例子中﹐你利用交疊了兩個FILD指令省下了兩個clock cycle。








